

NAG Library Routine Document

F08CJF (DORGRQ)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08CJF (DORGRQ) generates all or part of the real n by n orthogonal matrix Q from an RQ factorization computed by F08CHF (DGERQF).

2 Specification

```
SUBROUTINE F08CJF (M, N, K, A, LDA, TAU, WORK, LWORK, INFO)
  INTEGER          M, N, K, LDA, LWORK, INFO
  REAL (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *dorgrq*.

3 Description

F08CJF (DORGRQ) is intended to be used following a call to F08CHF (DGERQF), which performs an RQ factorization of a real matrix A and represents the orthogonal matrix Q as a product of k elementary reflectors of order n .

This routine may be used to generate Q explicitly as a square matrix, or to form only its trailing rows.

Usually Q is determined from the RQ factorization of a p by n matrix A with $p \leq n$. The whole of Q may be computed by:

```
CALL DORGRQ(N,N,P,A,LDA,TAU,WORK,LWORK,INFO)
```

(note that the matrix A must have at least n rows), or its trailing p rows as:

```
CALL DORGRQ(P,N,P,A,LDA,TAU,WORK,LWORK,INFO)
```

The rows of Q returned by the last call form an orthonormal basis for the space spanned by the rows of A ; thus F08CHF (DGERQF) followed by F08CJF (DORGRQ) can be used to orthogonalize the rows of A .

The information returned by F08CHF (DGERQF) also yields the RQ factorization of the trailing k rows of A , where $k < p$. The orthogonal matrix arising from this factorization can be computed by:

```
CALL DORGRQ(N,N,K,A,LDA,TAU,WORK,LWORK,INFO)
```

or its leading k columns by:

```
CALL DORGRQ(K,N,K,A,LDA,TAU,WORK,LWORK,INFO)
```

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: M – INTEGER *Input*
On entry: m , the number of rows of the matrix Q .
Constraint: $M \geq 0$.
- 2: N – INTEGER *Input*
On entry: n , the number of columns of the matrix Q .
Constraint: $N \geq M$.
- 3: K – INTEGER *Input*
On entry: k , the number of elementary reflectors whose product defines the matrix Q .
Constraint: $M \geq K \geq 0$.
- 4: A(LDA,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: details of the vectors which define the elementary reflectors, as returned by F08CHF (DGERQF).
On exit: the m by n matrix Q .
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08CJF (DORGRQ) is called.
Constraint: $LDA \geq \max(1, M)$.
- 6: TAU(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array TAU must be at least $\max(1, K)$.
On entry: $TAU(i)$ must contain the scalar factor of the elementary reflector H_i , as returned by F08CHF (DGERQF).
- 7: WORK(max(1,LWORK)) – REAL (KIND=nag_wp) array *Workspace*
On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimal performance.
- 8: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08CJF (DORGRQ) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Suggested value: for optimal performance, $LWORK \geq N \times nb$, where nb is the optimal **block size**.
Constraint: $LWORK \geq \max(1, M)$ or LWORK = -1.
- 9: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix Q differs from an exactly orthogonal matrix by a matrix E such that

$$\|E\|_2 = O\epsilon$$

and ϵ is the *machine precision*.

8 Parallelism and Performance

F08CJF (DORGRQ) is not threaded by NAG in any implementation.

F08CJF (DORGRQ) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $4mnk - 2(m+n)k^2 + \frac{4}{3}k^3$; when $m = k$ this becomes $\frac{2}{3}m^2(3n - m)$.

The complex analogue of this routine is F08CWF (ZUNGRQ).

10 Example

This example generates the first four rows of the matrix Q of the RQ factorization of A as returned by F08CHF (DGERQF), where

$$A = \begin{pmatrix} -0.57 & -1.93 & 2.30 & -1.93 & 0.15 & -0.02 \\ -1.28 & 1.08 & 0.24 & 0.64 & 0.30 & 1.03 \\ -0.39 & -0.31 & 0.40 & -0.66 & 0.15 & -1.43 \\ 0.25 & -2.14 & -0.35 & 0.08 & -2.13 & 0.50 \end{pmatrix}.$$

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

10.1 Program Text

```

Program f08cjfe

!      F08CJF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dgerqf, dorgrq, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                      :: i, ifail, info, lda, lwork, m, n

```

```

Character (26)                :: title
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,,:), tau(:), work(:)
! .. Executable Statements ..
Write (nout,*) 'F08CJF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*)
Read (nin,*) m, n
lda = m
lwork = nb*m
Allocate (a(lda,n),tau(n),work(lwork))

! Read A from data file

Read (nin,*)(a(i,1:n),i=1,m)

! Compute the RQ factorization of A
! The NAG name equivalent of dgerqf is f08chf
Call dgerqf(m,n,a,lda,tau,work,lwork,info)

! Form the leading M rows of Q explicitly
! The NAG name equivalent of dorgrq is f08cjf
Call dorgrq(m,n,m,a,lda,tau,work,lwork,info)

! Form the heading for X04CAF

Write (title,99999) m
Flush (nout)

! Print the leading M rows of Q

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04caf('General',' ',m,n,a,lda,title,ifail)

99999 Format ('The leading ',I4,' rows of Q')
End Program f08cjfe

```

10.2 Program Data

F08CJF Example Program Data

```

4      6                                :Values of M and N

-0.57  -1.93  2.30  -1.93  0.15  -0.02
-1.28  1.08  0.24  0.64  0.30  1.03
-0.39  -0.31  0.40  -0.66  0.15  -1.43
0.25  -2.14  -0.35  0.08  -2.13  0.50 :End of matrix A

```

10.3 Program Results

F08CJF Example Program Results

```

The leading      4 rows of Q
      1      2      3      4      5      6
1  -0.0833  0.2972 -0.6404  0.4461 -0.2938 -0.4575
2   0.9100 -0.1080 -0.2351 -0.1620  0.2022 -0.1946
3  -0.2202 -0.2706  0.2220 -0.3866  0.0015 -0.8243
4  -0.0809  0.6922  0.1132 -0.0259  0.6890 -0.1617

```
